GTTM1 - A FORTRAN PACKAGE FOR GRADIENT ANALYSIS IN POWER SYSTEMS

J.W. Bandler, M.A.El-Kady and J. Wojciechowski

SOS-83-30-L

December 1983

GTTM1 - A FORTRAN PACKAGE FOR GRADIENT ANALYSIS IN POWER SYSTEMS

J.W. Bandler, M.A. El-Kady and J. Wojciechowski

Abstract

This document contains a listing of the package GTTM1 for gradient evaluation in various power system applications. GTTM1 provides the values and the reduced gradients of a variety of functions frequently used in the optimal load flow and power system planning problems. The Tellegen theorem method is used with the aid of sparse matrix techniques for efficient gradient calculation.

The package GTTM1 has been developed for the CDC 170/815 system with the NOS 2.1 operating system and the Fortran Extended (FTN) version 4.8 compiler. The listing contains a total of 1334 lines (including 359 comments) constituting 33 subroutines.

This work was supported by the Natural Sciences and Engineering Research Council of Canada under Grants A7239, G0647 and A1708.

The authors are with the Simulation Optimization Systems Research Laboratory and the Department of Electrical and Computer Engineering, McMaster University, Hamilton, Canada L8S 4L7.

M.A. El-Kady is also with Ontario Hydro, Toronto, Canada.

J. Wojciechowski is on leave from the Institute of Electronics Fundamentals, Technical University of Warsaw, Warsaw, Poland.

I. INTRODUCTION

This document contains a Fortran listing of the package GTTM1 for gradient analysis in power systems. The user manual of the package together with illustrative examples is found in [1].

The package GTTM1 implements the Tellegen theorem method for gradient calculation [2,3]. It provides the values and the reduced gradients of a variety of functions frequently used in the optimal load flow and planning problems. These functions are: load bus voltage magnitude, bus voltage argument, generator bus reactive power, current magnitude of an admittance-type element, transmission power losses, slack bus active power, cost of generation. The allowed control variables are: input and output line shunt conductance and susceptance, transmission line conductance and susceptance, bus static load, load bus active and reactive powers, generator bus active power and voltage magnitude.

The package has been developed in Fortran IV for use on the CDC 170/815 system with the NOS 2.1 operating system. The package is available at McMaster University in the form of a library of binary relocatable subroutines. The library is in the group indirect file LIBGTM1 accessible under the charge RJWBAND. The package calls the subroutine SENSIT of the package TTM1 [3]. Subsequently subroutines DERIV, FORMTE, and PQ of TTM1 and also subroutines MA28A, MA28B and MA28C of the package MA28 [4] are called. Packages TTM1 and MA28 must thus be available when GTTM1 is used. This document includes neither of these packages.

The package GTTM1 has been modularized into 33 subroutines. It contains 1334 lines of which 359 are comments.

II. REFERENCES

[1] J.W. Bandler, M.A. El-Kady and J. Wojciechowski, "GTTM1 - A Fortran package for gradient analysis in power systems", Department of Electrical and Computer Engineering, McMaster University, Hamilton, Canada, Report SOS-83-30-U, 1983.

[2] J.W. Bandler and M.A. El-Kady, "Exact power network sensitivities via generalized complex branch modelling", Proc. IEEE Symp. CAS (Rome, Italy, May 1982), pp. 313-316.

[3] J.W. Bandler, M.A. El-Kady and J. Wojciechowski, "TTM1 - A Fortran implementation of the Tellegen theorem method to power system simulation and design", Department of Electrical and Computer Engineering, McMaster University, Hamilton, Canada, Report SOS-82-12-U2, 1983.

[4] I.A. Duff, "MA28 - A set of Fortran subroutines for sparse unsymmetric linear equations", Computer Science and Systems Division, AERE Harwell, Oxfordshire, England, Report R.8730, 1980.

III. LISTING OF THE PACKAGE GRAD

Subroutine	Number of lines (source text)	Number of words (compiled code)	Listing from page
WDADJ	35	104	5
GRAD1	33	263	6
GRAD2	34	266	7
GRAD3	41	456	8
GRAD4	46	475	9
GRAD5	46	507	10
GRAD6	40	452	11
GRAD7	40	430	12
GRAD8	45	576	13
PARDER3	68	420	14
PARDER4	62	347	16
PARDER5	65	367	17
PARDER6	56	261	18
PARDER7	64	407	19
PARDER8	84	610	20
RHST1	26	147	22
RHST2	27	161	23
RHST3G	23	136	24
RHST3S	39	202	25
RHST4	50	243	26
RHST5	51	256	27
RHST6	39	216	28
RHST7	39	173	29
RHST8	39	201	30
PDR3	32	117	31
PDR4	28	201	32
PDR5	26	142	33
PDR6	29	172	34
PDR7	32	117	35
PDR8	32	131	36
WRTGR	14	132	37
CF1	35	146	38
CF5	14	65	39

```
C
C
C
      SUBROUTINE WDADJ (WADJ,LADJ,NB,NTL)
C
C     SUBROUTINE WDADJ DISTRIBUTES WORKSPACE VECTOR WADJ INTO
C     A FAMILY OF SUBVECTORS USED FOR GRADIENT ANALYSIS
C
      REAL WADJ(LADJ)
      COMMON /MDADJ/ JT,JICN,JICT,JIRT,JIKEEP,JIW,JCS,JRHS,JPDR/FLAG/IFL
     1AG
C
      IFLAG=0
      N1=NB+NB
      NYT=NB+NTL+NTL
      NR=N1-2
      NZ=(7*NYT)/2
      LICN=3*NZ
      LIRN=NZ+NR
      JT=1
      JICN=JT+LICN
      JICT=JICN+LICN
      JIRT=JICT+NZ
      JIKEEP=JIRT+LIRN
      JIW=JIKEEP+5*NR
      JCS=JIW+8*NR
      JRHS=JCS+N1
      JPDR=JRHS+N1
      JMAX=JPDR+NB/2+NTL-1
      IF (JMAX.LE.LADJ) RETURN
      IFLAG=-1
      WRITE (6,10) JMAX
      RETURN
   10 FORMAT (///" WORKSPACE REQUIRED FOR GRADIENT ANALYSIS:",I4)
      END
```

```
C
C
C
      SUBROUTINE GRAD1 (LBINP,LBOUT,YT,JRYT,ICYT,V,BTYP,CCV,ICV,DF,F,WAD
     1J,NB,IND,NCV,MODE,IWRITE)
C
C     SUBROUTINE  GRAD1 CALCULATES VOLTAGE MODULUS OF THE INDTH BUS
C     AND ITS GRADIENTS W.R.T. CONTROL VARIABLES DEFINED BY VECTORS
C     CCV,ICV
C
      INTEGER LBINP(1),LBOUT(1),JRYT(1),ICYT(1),BTYP(1),CCV(1),ICV(1)
      REAL WADJ(1),DF(1)
      COMPLEX V(1),YT(1),VB
      COMMON /MDADJ/ JT,JICN,JICT,JIRT,JIKEEP,JIW,JCS,JRHS,JPDR/FLAG/IFL
     1AG
C
      IFLAG=0
      NAV=NB+NB
      VB=V(IND)
      CALL RHST1 (VB,IND,WADJ(JRHS),NAV,IWRITE)
      IVT=1
      CALL SENSIT (LBINP,LBOUT,YT,JRYT,ICYT,V,BTYP,WADJ(JCS),WADJ(JRHS),
     1CCV,ICV,WADJ(JPDR),DF,WADJ(JT),WADJ(JIRT),WADJ(JICT),WADJ(JICN),WA
     2DJ(JIKEEP),WADJ(JIW),NB,NCV,IVT,MODE,IFLAG,6,IWRITE-1)
      IF (IFLAG.LT.0) RETURN
      F=CABS(VB)
      IF (IWRITE.LT.3) RETURN
      WRITE (6,10) IND,F
      CALL WRTGR (CCV,ICV,DF,NCV)
      RETURN
   10 FORMAT (//" VOLTAGE MODULUS OF BUS:",I3/" VALUE OF FUNCTION",4X,E1
     13.7/" GRADIENTS")
      END
```

```
C
C
C
      SUBROUTINE GRAD2 (LBINP,LBOUT,YT,JRYT,ICYT,V,BTYP,CCV,ICV,DF,F,WAD
     1J,NB,IND,NCV,MODE,IWRITE)
C
C     SUBROUTINE GRAD2 CALCULATES VOLTAGE ARGUMENT OF THE INDTH BUS
C     AND ITS GRADIENTS W.R.T. CONTROL VARIABLES DEFINED BY VECTORS
C     CCV,ICV
C
      INTEGER LBINP(1),LBOUT(1),JRYT(1),ICYT(1),BTYP(1),CCV(1),ICV(1)
      REAL WADJ(1),DF(1)
      COMPLEX V(1),YT(1),VB
      COMMON /MDADJ/ JT,JICN,JICT,JIRT,JIKEEP,JIW,JCS,JRHS,JPDR/FLAG/IFL
     1AG
C
      IFLAG=0
      NAV=NB+NB
      VB=V(IND)
      ITYP=BTYP(IND)
      CALL RHST2 (VB,ITYP,IND,WADJ(JRHS),NAV,IWRITE)
      IVT=1
      CALL SENSIT (LBINP,LBOUT,YT,JRYT,ICYT,V,BTYP,WADJ(JCS),WADJ(JRHS),
     1CCV,ICV,WADJ(JPDR),DF,WADJ(JT),WADJ(JIRT),WADJ(JICT),WADJ(JICN),WA
     2DJ(JIKEEP),WADJ(JIW),NB,NCV,IVT,MODE,IFLAG,6,IWRITE-1)
      IF (IFLAG.LT.0) RETURN
      F=ATAN2(AIMAG(VB),REAL(VB))
      IF (IWRITE.LT.3) RETURN
      WRITE (6,10) IND,F
      CALL WRTGR (CCV,ICV,DF,NCV)
      RETURN
   10 FORMAT (//" VOLTAGE ARGUMENT OF BUS:",I3/" VALUE OF FUNCTION",4X,E
     113.7/" GRADIENTS")
      END
```

```
C
C
      SUBROUTINE GRAD3 (LBINP,LBOUT,YT,JRYT,ICYT,V,BTYP,CCV,ICV,DF,F,WAD
     1J,NB,IND,NCV,MODE,IWRITE)
C
C     SUBROUTINE  GRAD3 CALCULATES THE VALUE OF REACTIVE POWER AT THE
C     INDTH BUS AND ITS GRADIENTS W.R.T. CONTROL VARIABLES DEFINED BY
C     VECTORS CCV,ICV
C
      INTEGER LBINP(1),LBOUT(1),JRYT(1),ICYT(1),BTYP(1),CCV(1),ICV(1)
      REAL WADJ(1),DF(1)
      COMPLEX V(1),YT(1),VB,CF5
      COMMON /MDADJ/ JT,JICN,JICT,JIRT,JIKEEP,JIW,JCS,JRHS,JPDR/FLAG/IFL
     1AG
      Q(K)=AIMAG(CF5(V(K),YT(JRYT(K)),V,LO,ICYT(JRYT(K)),JRYT(K+1)-JRYT(
     1K),1))
C
      IFLAG=0
      NAV=NB+NB
      VB=V(IND)
      IF (IND.LT.NB) CALL RHST3G (VB,IND,WADJ(JRHS),NAV,IWRITE)
      IF (IND.EQ.NB) CALL RHST3S (YT,JRYT,ICYT,V,BTYP,WADJ(JRHS),NB,IWRI
     1TE)
      CALL PDR3 (LBINP,LBOUT,VB,CCV,ICV,WADJ(JPDR),IND,NCV,NDR)
      IVT=1
      CALL SENSIT (LBINP,LBOUT,YT,JRYT,ICYT,V,BTYP,WADJ(JCS),WADJ(JRHS),
     1CCV,ICV,WADJ(JPDR),DF,WADJ(JT),WADJ(JIRT),WADJ(JICT),WADJ(JICN),WA
     2DJ(JIKEEP),WADJ(JIW),NB,NCV,IVT,MODE,IFLAG,6,IWRITE-1)
      IF (IFLAG.LT.0) RETURN
      F=Q(IND)
      K=JRYT(IND)
      NYR=JRYT(IND+1)-K
      CALL PARDER3 (LBINP,LBOUT,YT(K),ICYT(K),V,CCV,ICV,DF,F,IND,NCV,NYR
     1)
      IF (IWRITE.LT.3) RETURN
      WRITE (6,10) IND,F
      CALL WRTGR (CCV,ICV,DF,NCV)
      RETURN
   10 FORMAT (//" REACTIVE POWER OF GENERATOR:",I3/" VALUE OF FUNCTION",
     14X,E13.7/" GRADIENTS")
      END
```

```
C
C
C
      SUBROUTINE GRAD4 (LBINP,LBOUT,YT,JRYT,ICYT,V,BTYP,CCV,ICV,DF,F,WAD
     1J,NB,YE,LTP,IND,NCV,MODE,IWRITE)
C
C     SUBROUTINE GRAD4 CALCULATES THE VALUE OF THE MODULUS OF CURRENT
C     OF THE INDTH ELEMENT AND ITS GRADIENTS W.R.T. VARIABLES DEFINED
C     BY VECTORS  CCV,ICV.  TYPE OF THE ELEMENT IS IDENTIFIED  BY THE
C     PARAMETER LTP
C
      INTEGER LBINP(1),LBOUT(1),JRYT(1),ICYT(1),BTYP(1),CCV(NCV),ICV(NCV
     1)
      REAL WADJ(1),DF(NCV)
      COMPLEX V(NB),YT(1),YE,CURR,CF,Z1,Z2,Z3
      COMMON /MDADJ/ JT,JICN,JICT,JIRT,JIKEEP,JIW,JCS,JRHS,JPDR/FLAG/IFL
     1AG
      CF(Z1,Z2,Z3)=Z1*(Z2-Z3)
C
      IFLAG=0
      NAV=NB+NB
      GO TO (10,20,30,40), LTP
   10 CURR=CF(YE,V(LBINP(IND)),V(LBOUT(IND)))
      GO TO 50
   20 CURR=CF(YE,V(LBINP(IND)),(0.,0.))
      GO TO 50
   30 CURR=CF(YE,V(LBOUT(IND)),(0.,0.))
      GO TO 50
   40 CURR=CF(YE,V(IND),(0.,0.))
   50 F=CABS(CURR)
      CALL RHST4 (LBINP,LBOUT,V,BTYP,WADJ(JRHS),YE,CURR,LTP,IND,NAV,IWRI
     1TE)
      CALL PDR4 (CCV,ICV,WADJ(JPDR),CURR,LTP,IND,NCV,NDR)
      IVT=1
      CALL SENSIT (LBINP,LBOUT,YT,JRYT,ICYT,V,BTYP,WADJ(JCS),WADJ(JRHS),
     1CCV,ICV,WADJ(JPDR),DF,WADJ(JT),WADJ(JIRT),WADJ(JICT),WADJ(JICN),WA
     2DJ(JIKEEP),WADJ(JIW),NB,NCV,IVT,MODE,IFLAG,6,IWRITE-1)
      IF (IFLAG.LT.0) RETURN
      CALL PARDER4 (LBINP,LBOUT,V,CCV,ICV,DF,F,YE,LTP,IND,NCV)
      IF (IWRITE.LT.3) RETURN
      WRITE (6,60) IND,LTP,F
      CALL WRTGR (CCV,ICV,DF,NCV)
      RETURN
   60 FORMAT (//" MODULUS OF CURRENT. ELEMENT INDEX",I3,",   ELEMENT TYPE
     1",I2,"."/" VALUE OF FUNCTION",4X,E13.7/" GRADIENTS")
      END
```

```
C
C
C
      SUBROUTINE GRAD5 (LBINP,LBOUT,YT,JRYT,ICYT,V,BTYP,CCV,ICV,DF,F,WAD
     1J,NB,YE,LTP,IND,NCV,MODE,IWRITE)
C
C     SUBROUTINE GRAD5  CALCULATES THE VALUE OF THE POWER LOSSES OF
C     THE INDTH ELEMENT AND ITS GRADIENTS W.R.T. VARIABLES  DEFINED
C     BY VECTORS CCV, ICV. TYPE OF THE ELEMENT IS IDENTIFIED BY THE
C     PARAMETER LTP
C
      INTEGER LBINP(1),LBOUT(1),JRYT(1),ICYT(1),BTYP(1),CCV(NCV),ICV(NCV
     1)
      REAL WADJ(1),DF(NCV)
      COMPLEX V(NB),YT(1),YE,CURR,CF,Z1,Z2,Z3
      COMMON /MDADJ/ JT,JICN,JICT,JIRT,JIKEEP,JIW,JCS,JRHS,JPDR/FLAG/IFL
     1AG
      CF(Z1,Z2,Z3)=Z1*(Z2-Z3)
C
      IFLAG=0
      NAV=NB+NB
      GO TO (10,20,30,40), LTP
   10 CURR=CF(YE,V(LBINP(IND)),V(LBOUT(IND)))
      GO TO 50
   20 CURR=CF(YE,V(LBINP(IND)),(0.,0.))
      GO TO 50
   30 CURR=CF(YE,V(LBOUT(IND)),(0.,0.))
      GO TO 50
   40 CURR=CF(YE,V(IND),(0.,0.))
   50 F=CABS(CURR/YE)
      F=F*F*REAL(YE)
      CALL RHST5 (LBINP,LBOUT,V,BTYP,WADJ(JRHS),YE,LTP,IND,NAV,IWRITE)
      CALL PDR5 (CCV,ICV,WADJ(JPDR),CURR,F,LTP,IND,NCV,NDR)
      IVT=1
      CALL SENSIT (LBINP,LBOUT,YT,JRYT,ICYT,V,BTYP,WADJ(JCS),WADJ(JRHS),
     1CCV,ICV,WADJ(JPDR),DF,WADJ(JT),WADJ(JIRT),WADJ(JICT),WADJ(JICN),WA
     2DJ(JIKEEP),WADJ(JIW),NB,NCV,IVT,MODE,IFLAG,6,IWRITE-1)
      IF (IFLAG.LT.0) RETURN
      CALL PARDER5 (LBINP,LBOUT,V,CCV,ICV,DF,F,YE,LTP,IND,NCV)
      IF (IWRITE.LT.3) RETURN
      WRITE (6,60) IND,LTP,F
      CALL WRTGR (CCV,ICV,DF,NCV)
      RETURN
   60 FORMAT (//" MODULUS OF CURRENT. ELEMENT INDEX",I3,",   ELEMENT TYPE
     1",I2,"."/" VALUE OF FUNCTION",4X,E13.7/" GRADIENTS")
      END
```

```
C
C
C
      SUBROUTINE GRAD6 (LBINP,LBOUT,LINPG,LINPB,LG,LB,YT,JRYT,ICYT,V,BTY
     1P,CCV,ICV,DF,F,WADJ,NB,NCV,MODE,IWRITE)
C
C     SUBROUTINE GRAD6 CALCULATES THE VALUE OF TRANSMISSION POWER LOSSES
C     AND ITS GRADIENTS W.R.T. CONTROL VARIABLES DEFINED BY VECTORS CCV,
C     ICV
C
      INTEGER LBINP(1),LBOUT(1),JRYT(1),ICYT(1),BTYP(1),CCV(1),ICV(1)
      REAL LINPG(1),LINPB(1),LG(1),LB(1),WADJ(1),DF(1)
      COMPLEX V(1),YT(1),CF5
      COMMON /MDADJ/ JT,JICN,JICT,JIRT,JIKEEP,JIW,JCS,JRHS,JPDR/FLAG/IFL
     1AG
      P(K)=REAL(CF5(V(K),YT(JRYT(K)),V,LO,ICYT(JRYT(K)),JRYT(K+1)-JRYT(K
     1),1))
C
      IFLAG=0
      NAV=NB+NB
      CALL RHST6 (YT,JRYT,ICYT,V,BTYP,WADJ(JRHS),NAV,IWRITE)
      CALL PDR6 (LBINP,LBOUT,LINPG,LINPB,LG,LB,V,CCV,ICV,WADJ(JPDR),NCV,
     1NDR)
      IVT=1
      CALL SENSIT (LBINP,LBOUT,YT,JRYT,ICYT,V,BTYP,WADJ(JCS),WADJ(JRHS),
     1CCV,ICV,WADJ(JPDR),DF,WADJ(JT),WADJ(JIRT),WADJ(JICT),WADJ(JICN),WA
     2DJ(JIKEEP),WADJ(JIW),NB,NCV,IVT,MODE,IFLAG,6,IWRITE-1)
      IF (IFLAG.LT.0) RETURN
      F=0.
      DO 10 I=1,NB
      F=F+P(I)
   10 CONTINUE
      CALL PARDER6 (LBINP,LBOUT,YT,JRYT,ICYT,V,CCV,ICV,DF,NCV)
      IF (IWRITE.LT.3) RETURN
      WRITE (6,20) F
      CALL WRTGR (CCV,ICV,DF,NCV)
      RETURN
   20 FORMAT (//" TRANSMISSION POWER LOSSES"/" VALUE OF FUNCTION",4X,E13
     1.7/" GRADIENTS")
      END
```

```
C
C
C
      SUBROUTINE GRAD7 (LBINP,LBOUT,YT,JRYT,ICYT,V,BTYP,CCV,ICV,DF,F,WAD
     1J,NB,IND,NCV,MODE,IWRITE)
C
C     SUBROUTINE GRAD7 CALCULATES THE VALUE OF ACTIVE POWER AT THE INDTH
C     BUS AND ITS  GRADIENTS W.R.T. CONTROL VARIABLES DEFINED BY VECTORS
C     CCV,ICV
C
      INTEGER LBINP(1),LBOUT(1),JRYT(1),ICYT(1),BTYP(1),CCV(1),ICV(1)
      REAL WADJ(1),DF(1)
      COMPLEX V(1),YT(1),VB,CF5
      COMMON /MDADJ/ JT,JICN,JICT,JIRT,JIKEEP,JIW,JCS,JRHS,JPDR/FLAG/IFL
     1AG
      P(K)=REAL(CF5(V(K),YT(JRYT(K)),V,LO,ICYT(JRYT(K)),JRYT(K+1)-JRYT(K
     1),1))
C
      IFLAG=0
      NAV=NB+NB
      VB=V(IND)
      CALL RHST7 (YT,JRYT,ICYT,V,BTYP,WADJ(JRHS),NB,IWRITE)
      CALL PDR7 (LBINP,LBOUT,VB,CCV,ICV,WADJ(JPDR),IND,NCV,NDR)
      IVT=1
      CALL SENSIT (LBINP,LBOUT,YT,JRYT,ICYT,V,BTYP,WADJ(JCS),WADJ(JRHS),
     1CCV,ICV,WADJ(JPDR),DF,WADJ(JT),WADJ(JIRT),WADJ(JICT),WADJ(JICN),WA
     2DJ(JIKEEP),WADJ(JIW),NB,NCV,IVT,MODE,IFLAG,6,IWRITE-1)
      IF (IFLAG.LT.0) RETURN
      F=P(IND)
      K=JRYT(IND)
      NYR=JRYT(IND+1)-K
      CALL PARDER7 (LBINP,LBOUT,YT(K),ICYT(K),V,CCV,ICV,DF,F,IND,NCV,NYR
     1)
      IF (IWRITE.LT.3) RETURN
      WRITE (6,10) IND,F
      CALL WRTGR (CCV,ICV,DF,NCV)
      RETURN
   10 FORMAT (//" ACTIVE POWER OF GENERATOR:",I3/" VALUE OF FUNCTION",4X
     1,E13.7/" GRADIENTS")
      END
```

```
C
C
C
      SUBROUTINE GRAD8 (LBINP,LBOUT,YT,JRYT,ICYT,V,BTYP,CCV,ICV,C,D,DF,F
     1,WADJ,NB,NCV,MODE,IWRITE)
C
C     SUBROUTINE  GRAD8  DETERMINES THE COST OF GENERATION  AND ITS
C     GRADIENTS W.R.T. CONTROL VARIABLES DEFINED BY VECTORS CCV,ICV
C
      INTEGER LBINP(1),LBOUT(1),JRYT(1),ICYT(1),BTYP(1),CCV(1),ICV(1)
      REAL C(1),D(1),WADJ(1),DF(1)
      COMPLEX V(1),YT(1),CF5
      COMMON /MDADJ/ JT,JICN,JICT,JIRT,JIKEEP,JIW,JCS,JRHS,JPDR/FLAG/IFL
     1AG
      P(K)=REAL(CF5(V(K),YT(JRYT(K)),V,LO,ICYT(JRYT(K)),JRYT(K+1)-JRYT(K
     1),1))
C
      IFLAG=0
      NAV=NB+NB
      PN=P(NB)
      J=0
      F=0.
      DO 10 I=1,NB
      IF (BTYP(I).EQ.0) GO TO 10
      J=J+1
      F=F+(C(J)+D(J)*P(I))*P(I)
   10 CONTINUE
      CALL RHST8 (YT,JRYT,ICYT,V,BTYP,WADJ(JRHS),C(J),D(J),PN,NAV,IWRITE
     1)
      CALL PDR8 (LBINP,LBOUT,V(NB),CCV,ICV,WADJ(JPDR),C(J),D(J),PN,NB,NC
     1V,NDR)
      IVT=1
      CALL SENSIT (LBINP,LBOUT,YT,JRYT,ICYT,V,BTYP,WADJ(JCS),WADJ(JRHS),
     1CCV,ICV,WADJ(JPDR),DF,WADJ(JT),WADJ(JIRT),WADJ(JICT),WADJ(JICN),WA
     2DJ(JIKEEP),WADJ(JIW),NB,NCV,IVT,MODE,IFLAG,6,IWRITE-1)
      IF (IFLAG.LT.0) RETURN
      CALL PARDER8 (LBINP,LBOUT,YT,JRYT,ICYT,V,BTYP,CCV,ICV,C,D,DF,NB,J,
     1NCV)
      IF (IWRITE.LT.3) RETURN
      WRITE (6,20) F
      CALL WRTGR (CCV,ICV,DF,NCV)
      RETURN
   20 FORMAT (//" COST OF GENERATION"/" VALUE OF FUNCTION",4X,E13.7/" GR
     1ADIENTS")
      END
```

```
C
C
C
      SUBROUTINE PARDER3 (LBINP,LBOUT,YR,ICYR,V,CCV,ICV,DF,F,INB,NCV,NYR
     1)
C
C     SUBROUTINE PARDER3 CALCULATES PARTIAL DERIVATIVES OF THE INBTH BUS
C     REACTIVE POWER W.R.T. CONTROL VARIABLES DEFINED BY VECTORS CCV,ICV
C
      INTEGER LBINP(1),LBOUT(1),ICYR(NYR),CCV(NCV),ICV(NCV)
      REAL DF(NCV)
      COMPLEX V(1),YR(NYR),VB,VR,CF5
C
      VB=V(INB)
      RV=CABS(VB)
      DO 110 J=1,NCV
      L=CCV(J)
      M=ICV(J)
C
C           LINPG LINPB LG LB LOUTG LOUTB P  Q  MODV ARGV BSTL
      GO TO (110,10,20,20,110,50,110,110,60,60,100), L
C
C     PARTIAL DERIVATIVE W.R.T. INPUT SHUNT SUSCEPTANCE
C
   10 IF (LBINP(M).EQ.INB) DF(J)=DF(J)-RV*RV
      GO TO 110
C
C     PARTIAL DERIVATIVE W.R.T. TRANSMISSION LINE ADMITTANCE
C
   20 IF (LBINP(M).NE.INB) GO TO 30
      K=LBOUT(M)
      GO TO 40
   30 IF (LBOUT(M).NE.INB) GO TO 110
      K=LBINP(M)
   40 IF (L.EQ.3) DF(J)=DF(J)+AIMAG(CONJG(VB)*V(K))
      IF (L.EQ.4) DF(J)=DF(J)+REAL(CONJG(VB)*V(K))-RV*RV
      GO TO 110
C
C     PARTIAL DERIVATIVE W.R.T. OUTPUT SHUNT SUSCEPTANCE
C
   50 IF (LBOUT(M).EQ.INB) DF(J)=DF(J)-RV*RV
      GO TO 110
C
C     PARTIAL DERIVATIVES W.R.T. BUS VOLTAGE MODULUS AND ARGUMENT
C
   60 IF (M.NE.INB) GO TO 70
      IF (L.EQ.9) DF(J)=DF(J)-RV*AIMAG(YR(1))+F/RV
      IF (L.EQ.10) DF(J)=DF(J)-RV*RV*REAL(YR(1))+REAL(CF5(VB,YR,V,K,ICYR
     1,NYR,1))
      GO TO 110
   70 IND=0
      DO 80 K=2,NYR
      IF (ICYR(K).NE.M) GO TO 80
      IND=K
      GO TO 90
   80 CONTINUE
   90 IF (IND.EQ.0) GO TO 110
      VR=V(ICYR(IND))
      IF (L.EQ.9) DF(J)=DF(J)+AIMAG(VB*CONJG(YR(IND)*VR))/CABS(VR)
      IF (L.EQ.10) DF(J)=DF(J)-REAL(VB*CONJG(YR(IND)*VR))
      GO TO 110
C
C     PARTIAL DERIVATIVE W.R.T. BUS STATIC LOAD
C
  100 IF (M.EQ.INB) DF(J)=DF(J)-RV*RV
```

```
110 CONTINUE                                                    J  66
    RETURN                                                      J  67
    END                                                         J  68
```

```
C
C
C
      SUBROUTINE PARDER4 (LBINP,LBOUT,V,CCV,ICV,DF,F,YE,LTP,INL,NCV)
C
C     SUBROUTINE  PARDER4 CALCULATES PARTIAL DERIVATIVES OF THE MODULUS
C     OF CURRENT OF THE INLTH  ELEMENT W.R.T. CONTROL VARIABLES DEFINED
C     BY VECTORS CCV,ICV.  TYPE OF THE ELEMENT IS IDENTIFIED BY THE
C     PARAMETER LTP.
C
      INTEGER LBINP(1),LBOUT(1),CCV(NCV),ICV(NCV)
      REAL DF(NCV)
      COMPLEX V(1),YE,V1,V2
C
      DO 110 J=1,NCV
      L=CCV(J)
      M=ICV(J)
C
C           LINPG LINPB LG LB LOUTG LOUTB  P    Q MODV ARGV BSTL
      GO TO (10,20,10,20,10,20,110,110,30,30,20), L
C
C     PARTIAL DERIVATIVE W.R.T. CONDUCTANCE
C
   10 IF (M.EQ.INL) DF(J)=DF(J)+REAL(1./YE)*F
      GO TO 110
C
C     PARTIAL DERIVATIVE W.R.T. SUSCEPTANCE
C
   20 IF (M.EQ.INL) DF(J)=DF(J)-AIMAG(1./YE)*F
      GO TO 110
C
C     PARTIAL DERIVATIVE W.R.T. BUS VOLTAGE MODULUS AND ARGUMENT
C
   30 GO TO (40,60,70,80), LTP
   40 IF (LBINP(INL).NE.M) GO TO 50
      V1=V(LBINP(INL))
      V2=V1*CONJG(V(LBOUT(INL)))
      GO TO 90
   50 IF (LBOUT(INL).NE.M) GO TO 110
      V1=V(LBOUT(INL))
      V2=V1*CONJG(V(LBINP(INL)))
      GO TO 90
   60 IF (LBINP(INL).NE.M) GO TO 110
      V1=V(LBINP(INL))
      V2=(0.,0.)
      GO TO 90
   70 IF (LBOUT(INL).NE.M) GO TO 110
      V1=V(LBOUT(INL))
      V2=(0.,0.)
      GO TO 90
   80 IF (INL.NE.M) GO TO 110
      V1=V(INL)
      V2=(0.,0.)
   90 R1=REAL(YE*CONJG(YE))/F
      IF (L.EQ.10) GO TO 100
      R2=CABS(V1)
      DF(J)=DF(J)+(R2-REAL(V2)/R2)*R1
      GO TO 110
  100 DF(J)=DF(J)+AIMAG(V2)*R1
  110 CONTINUE
      RETURN
      END
```

```
C
C
C
      SUBROUTINE PARDER5 (LBINP,LBOUT,V,CCV,ICV,DF,F,YE,LTP,INL,NCV)
C
C     SUBROUTINE PARDER5 CALCULATES PARTIAL DERIVATIVES OF THE POWER
C     LOSSES OF THE INLTH  ELEMENT W.R.T. CONTROL  VARIABLES DEFINED
C     BY VECTORS CCV,  ICV.  TYPE OF THE ELEMENT IS IDENTIFIED BY THE
C     PARAMETER LTP.
C
      INTEGER LBINP(1),LBOUT(1),CCV(NCV),ICV(NCV)
      REAL DF(NCV)
      COMPLEX V(1),YE,V1,V2
C
      DO 150 J=1,NCV
      L=CCV(J)
      M=ICV(J)
C
C           LINPG LINPB LG  LB LOUTG LOUTB  P    Q MODV ARGV BSTL
      GO TO (10,150,10,150,10,150,150,150,70,70,150), L
C
C     PARTIAL DERIVATIVE W.R.T. LINE PARAMETERS
C
   10 IF (M.NE.INL) GO TO 150
      GO TO (20,30,40,50), LTP
   20 R1=CABS(V(LBINP(INL))-V(LBOUT(INL)))
      GO TO 60
   30 R1=CABS(V(LBINP(INL)))
      GO TO 60
   40 R1=CABS(V(LBOUT(INL)))
      GO TO 60
   50 R1=CABS(V(INL))
   60 DF(J)=DF(J)+R1*R1
      GO TO 150
C
C     PARTIAL DERIVATIVE W.R.T. BUS VOLTAGE MODULUS AND ARGUMENT
C
   70 GO TO (80,100,110,120), LTP
   80 IF (LBINP(INL).NE.M) GO TO 90
      V1=V(LBINP(INL))
      V2=V1*CONJG(V(LBOUT(INL)))
      GO TO 130
   90 IF (LBOUT(INL).NE.M) GO TO 150
      V1=V(LBOUT(INL))
      V2=V1*CONJG(V(LBINP(INL)))
      GO TO 130
  100 IF (LBINP(INL).NE.M) GO TO 150
      V1=V(LBINP(INL))
      V2=(0.,0.)
      GO TO 130
  110 IF (LBOUT(INL).NE.M) GO TO 150
      V1=V(LBOUT(INL))
      V2=(0.,0.)
      GO TO 130
  120 IF (INL.NE.M) GO TO 150
      V1=V(INL)
      V2=(0.,0.)
  130 IF (L.EQ.10) GO TO 140
      R2=CABS(V1)
      DF(J)=DF(J)+2.*REAL(YE)*(R2-REAL(V2)/R2)
      GO TO 150
  140 DF(J)=DF(J)+2.*AIMAG(V2)*REAL(YE)
  150 CONTINUE
      RETURN
      END
```

```
C
C
C
      SUBROUTINE PARDER6 (LBINP,LBOUT,YT,JRYT,ICYT,V,CCV,ICV,DF,NCV)
C
C     SUBROUTINE PARDER6 CALCULATES PARTIAL DERIVATIVES OF TRANSMISSION
C     POWER LOSSES W.R.T. CONTROL VARIABLES  DEFINED BY VECTORS CCV,ICV
C
      INTEGER LBINP(1),JRYT(1),LBOUT(1),ICYT(1),CCV(NCV),ICV(NCV)
      REAL DF(NCV)
      COMPLEX YT(1),V(1),VB
C
      DO 90 J=1,NCV
      L=CCV(J)
      M=ICV(J)
C
C           LINPG LINPB LG LB LOUTG LOUTB  P    Q MODV ARGV BSTL
      GO TO (10,90,20,90,30,90,90,90,50,50,90), L
C
C     PARTIAL DERIVATIVE W.R.T. LINE INPUT SHUNT CONDUCTANCE
C
   10 VB=V(LBINP(M))
      GO TO 40
C
C     PARTIAL DERIVATIVE W.R.T. LINE CONDUCTANCE
C
   20 VB=V(LBINP(M))-V(LBOUT(M))
      GO TO 40
C
C     PARTIAL DERIVATIVE W.R.T. LINE OUTPUT SHUNT CONDUCTANCE
C
   30 VB=V(LBOUT(M))
   40 DF(J)=DF(J)+REAL(VB*CONJG(VB))
      GO TO 90
C
C     PARTIAL DERIVATIVES W.R.T. BUS VOLTAGE MODULUS AND ARGUMENT
C
   50 K1=JRYT(M)+1
      K2=JRYT(M+1)-1
      R=0.
      VB=V(M)
      IF (L.EQ.10) GO TO 70
      VM=CABS(VB)
      DO 60 K=K1,K2
      R=R+REAL(YT(K))*REAL(VB*CONJG(V(ICYT(K))))
   60 CONTINUE
      R=VM*REAL(YT(K1-1))+R/VM
      DF(J)=DF(J)+R+R
      GO TO 90
   70 DO 80 K=K1,K2
      R=R-REAL(YT(K))*AIMAG(VB*CONJG(V(ICYT(K))))
   80 CONTINUE
      DF(J)=DF(J)+R+R
   90 CONTINUE
      RETURN
      END
```

```
C
C
C
      SUBROUTINE PARDER7 (LBINP,LBOUT,YR,ICYR,V,CCV,ICV,DF,F,INB,NCV,NYR
     1)
C
C     SUBROUTINE PARDER7 CALCULATES PARTIAL DERIVATIVES OF THE INBTH
C     BUS ACTIVE POWER  W.R.T. CONTROL VARIABLES  DEFINED BY VECTORS
C     CCV,ICV
C
      INTEGER LBINP(1),LBOUT(1),ICYR(NYR),CCV(NCV),ICV(NCV)
      REAL DF(NCV)
      COMPLEX V(1),YR(NYR),VB,VR,CF5
C
      VB=V(INB)
      RV=CABS(VB)
      DO 100 J=1,NCV
      L=CCV(J)
      M=ICV(J)
C
C             LINPG LINPB LG LB LOUTG LOUTB P    Q  MODV ARGV BSTL
      GO TO (10,100,20,20,50,100,100,100,60,60,100), L
C
C     PARTIAL DERIVATIVE W.R.T. LINE INPUT SHUNT CONDUCTANCE
C
   10 IF (LBINP(M).EQ.INB) DF(J)=DF(J)+RV*RV
      GO TO 100
C
C     PARTIAL DERIVATIVE W.R.T. TRANSMISSION LINE ADMITTANCE
C
   20 IF (LBINP(M).NE.INB) GO TO 30
      K=LBOUT(M)
      GO TO 40
   30 IF (LBOUT(M).NE.INB) GO TO 100
      K=LBINP(M)
   40 IF (L.EQ.3) DF(J)=DF(J)-REAL(CONJG(VB)*V(K))+RV*RV
      IF (L.EQ.4) DF(J)=DF(J)+AIMAG(CONJG(VB)*V(K))
      GO TO 100
C
C     PARTIAL DERIVATIVE W.R.T. LINE OUTPUT SHUNT CONDUCTANCE
C
   50 IF (LBOUT(M).EQ.INB) DF(J)=DF(J)+RV*RV
      GO TO 100
C
C     PARTIAL DERIVATIVES W.R.T. BUS VOLTAGE MODULUS AND ARGUMENT
C
   60 IF (M.NE.INB) GO TO 70
      IF (L.EQ.9) DF(J)=DF(J)+RV*REAL(YR(1))+F/RV
      IF (L.EQ.10) DF(J)=DF(J)-RV*RV*AIMAG(YR(1))-AIMAG(CF5(VB,YR,V,K,IC
     1YR,NYR,1))
      GO TO 100
   70 IND=0
      DO 80 K=2,NYR
      IF (ICYR(K).NE.M) GO TO 80
      IND=K
      GO TO 90
   80 CONTINUE
   90 IF (IND.EQ.0) GO TO 100
      VR=V(ICYR(IND))
      IF (L.EQ.9) DF(J)=DF(J)+REAL(VB*CONJG(YR(IND)*VR))/CABS(VR)
      IF (L.EQ.10) DF(J)=DF(J)+AIMAG(VB*CONJG(YR(IND)*VR))
  100 CONTINUE
      RETURN
      END
```

```
C
C
C
      SUBROUTINE PARDER8 (LBINP,LBOUT,YT,JRYT,ICYT,V,BTYP,CCV,ICV,C,D,DF
     1,NB,NG,NCV)
C
C     SUBROUTINE PARDER8 DETERMINES PARTIAL DERIVATIVES OF THE COST OF
C     GENERATION  W.R.T. CONTROL VARIABLES DEFINED BY VECTORS CCV, ICV
C
      INTEGER LBINP(1),LBOUT(1),JRYT(1),ICYT(1),BTYP(1),CCV(NCV),ICV(NCV
     1)
      REAL C(NG),D(NG),DF(1)
      COMPLEX V(1),YT(1),VB,VR,CF5
      DER(A1,A2,A3,X)=(A1+2.*A2*A3*X)*X
      P(K)=REAL(CF5(V(K),YT(JRYT(K)),V,LO,ICYT(JRYT(K)),JRYT(K+1)-JRYT(K
     1),1))
      Q(K)=AIMAG(CF5(V(K),YT(JRYT(K)),V,LO,ICYT(JRYT(K)),JRYT(K+1)-JRYT(
     1K),1))
C
      VB=V(NB)
      RV=CABS(VB)
      RV2=RV*RV
      CN=C(NG)
      DN=D(NG)
      PN=P(NB)
      K1=JRYT(NB)+1
      K2=JRYT(NB+1)-1
      DO 130 J=1,NCV
      L=CCV(J)
      M=ICV(J)
C
C          LINPG LINPB LG LB LOUTG LOUTB P    Q  MODV ARGV BSTL
      GO TO (10,130,20,20,50,130,60,130,90,90,130), L
C
C     PARTIAL DERIVATIVE W.R.T. LINE INPUT SHUNT CONDUCTANCE
C
   10 IF (LBINP(M).EQ.NB) DF(J)=DF(J)+DER(CN,DN,PN,RV2)
      GO TO 130
C
C     PARTIAL DERIVATIVE W.R.T. TRANSMISSION LINE ADMITTANCE
C
   20 IF (LBINP(M).NE.NB) GO TO 30
      K=LBOUT(M)
      GO TO 40
   30 IF (LBOUT(M).NE.NB) GO TO 130
      K=LBINP(M)
   40 IF (L.EQ.3) DF(J)=DF(J)+DER(CN,DN,PN,RV2-REAL(CONJG(VB)*V(K)))
      IF (L.EQ.4) DF(J)=DF(J)+DER(CN,DN,PN,AIMAG(CONJG(VB)*V(K)))
      GO TO 130
C
C     PARTIAL DERIVATIVE W.R.T. LINE OUTPUT SHUNT CONDUCTANCE
C
   50 IF (LBOUT(M).EQ.NB) DF(J)=DF(J)+DER(CN,DN,PN,RV2)
      GO TO 130
   60 I=0
      N1=NB-1
      DO 70 IG=1,N1
      IF (BTYP(IG).EQ.0) GO TO 70
      I=I+1
      IF (IG.EQ.M) GO TO 80
   70 CONTINUE
   80 DF(J)=DF(J)+C(I)+2.*D(I)*P(M)
      GO TO 130
C
C     PARTIAL DERIVATIVES W.R.T. BUS VOLTAGE MODULUS AND ARGUMENT
```

```
C
   90 IF (M.NE.NB) GO TO 100
      IF (L.EQ.9) DF(J)=DF(J)+DER(CN,DN,PN,RV*REAL(YT(K1-1))+PN/RV)
      IF (L.EQ.10) DF(J)=DF(J)+DER(CN,DN,PN,-Q(NB)-RV2*AIMAG(YT(K1-1)))
      GO TO 130
  100 IND=0
      DO 110 K=K1,K2
      IF (ICYT(K).NE.M) GO TO 110
      IND=K
      GO TO 120
  110 CONTINUE
  120 IF (IND.EQ.0) GO TO 130
      VR=V(ICYT(IND))
      IF (L.EQ.9) DF(J)=DF(J)+DER(CN,DN,PN,REAL(VB*CONJG(YT(IND)*VR))/CA
     1BS(VR))
      IF (L.EQ.10) DF(J)=DF(J)+DER(CN,DN,PN,AIMAG(VB*CONJG(YT(IND)*VR)))
  130 CONTINUE
      RETURN
      END
```

```
C
C
C
      SUBROUTINE RHST1 (VB,INB,RHS,NAV,IWRITE)
C
C     SUBROUTINE  RHST1 FORMS THE RIGHT HAND SIDE VECTOR RHS OF THE
C     ADJOINT EQUATIONS FOR THE MODULUS OF VOLTAGE OF THE INBTH BUS
C
      REAL RHS(NAV)
      COMPLEX VB,VA
C
      DO 10 J=1,NAV
      RHS(J)=0.
   10 CONTINUE
      J2=INB+INB
      J1=J2-1
      VA=CABS(VB)/VB
      RHS(J1)=REAL(VA)
      RHS(J2)=AIMAG(VA)
      IF (IWRITE.LT.4) GO TO 20
      WRITE (6,30)
      WRITE (6,40) (M,RHS(M),M=1,NAV)
   20 RETURN
   30 FORMAT (//" RHS VECTOR OF ADJOINT EQUATIONS"/)
   40 FORMAT (5(2X,I4,":",E13.7))
      END
```

```
C
C
C
      SUBROUTINE RHST2 (VB,ITYP,INB,RHS,NAV,IWRITE)
C
C     SUBROUTINE  RHST2 FORMS  THE RIGHT HAND SIDE VECTOR RHS OF THE
C     ADJOINT OF VOLTAGE EQUATIONS FOR THE ARGUMENT OF THE INBTH BUS
C
      REAL RHS(NAV)
      COMPLEX VA,VB
C
      DO 10 J=1,NAV
      RHS(J)=0.
   10 CONTINUE
      J2=INB+INB
      J1=J2-1
      VA=(-1.,0.)
      IF (ITYP.EQ.0) VA=(0.,-1.)/VB
      RHS(J1)=REAL(VA)
      RHS(J2)=AIMAG(VA)
      IF (IWRITE.LT.4) GO TO 20
      WRITE (6,30)
      WRITE (6,40) (M,RHS(M),M=1,NAV)
   20 RETURN
   30 FORMAT (//" RHS VECTOR RHS OF ADJOINT EQUATIONS"/)
   40 FORMAT (5(2X,I4,":",E13.7))
      END
```

```
C
C
C
      SUBROUTINE RHST3G (VB,INB,RHS,NAV,IWRITE)
C
C
C     SUBROUTINE  RHST3  FORMS THE RIGHT HAND SIDE  VECTOR RHS OF THE
C     ADJOINT EQUATIONS FOR THE REACTIVE POWER OF THE INBTH GENERATOR
C
      REAL RHS(NAV)
      COMPLEX VB
C
      DO 10 J=1,NAV
      RHS(J)=0.
   10 CONTINUE
      RHS(INB+INB)=-REAL(VB*CONJG(VB))
      IF (IWRITE.LT.4) GO TO 20
      WRITE (6,30)
      WRITE (6,40) (M,RHS(M),M=1,NAV)
   20 RETURN
   30 FORMAT (//" RHS VECTOR OF ADJOINT EQUATIONS"/)
   40 FORMAT (5(2X,I4,":",E13.7))
      END
```

```
C
C
C
      SUBROUTINE RHST3S (YT,JRYT,ICYT,V,BTYP,RHS,NB,IWRITE)
C
C     SUBROUTINE RHST3S FORMS THE RIGHT HAND SIDE VECTOR RHS OF THE
C     ADJOINT EQUATIONS FOR THE REACTIVE POWER OF THE SLACK BUS
C
      INTEGER JRYT(1),ICYT(1),BTYP(NB)
      REAL RHS(1)
      COMPLEX YT(1),V(NB),Z,ZZ
C
      NAV=NB+NB
      NR=NAV-2
      DO 10 J=1,NR
      RHS(J)=0.0
   10 CONTINUE
      Z=CONJG(V(NB))
      RHS(NAV-1)=AIMAG(Z)
      RHS(NAV)=-REAL(Z)
      K1=JRYT(NB)+1
      K2=JRYT(NB+1)-1
      DO 30 L=K1,K2
      IBUS=ICYT(L)
      LL=IBUS+IBUS
      ZZ=YT(L)*Z*(0.,1.)
      IF (BTYP(IBUS).EQ.1) GO TO 20
      RHS(LL-1)=REAL(ZZ)
      RHS(LL)=AIMAG(ZZ)
      GO TO 30
   20 RHS(LL-1)=AIMAG(ZZ*V(IBUS))
   30 CONTINUE
      IF (IWRITE.LT.4) GO TO 40
      WRITE (6,50)
      WRITE (6,60) (M,RHS(M),M=1,NAV)
   40 RETURN
   50 FORMAT (//" RHS VECTOR OF ADJOINT EQUATIONS"/)
   60 FORMAT (5(2X,I4,": ",E13.7))
      END
```

```
C
C
C
      SUBROUTINE RHST4 (LBINP,LBOUT,V,BTYP,RHS,YE,CURR,LTP,INL,NAV,IWRIT
     1E)
C
C     SUBROUTINE RHST4 FORMULATES RHS VECTOR OF THE ADJOINT EQUATIONS
C     FOR THE MODULUS  OF CURRENT  OF THE INLTH ELEMENT.  TYPE OF THE
C     ELEMENT IS IDENTIFIED BY THE PARAMETER LTP.
C
      INTEGER LBINP(1),LBOUT(1),BTYP(1)
      REAL RHS(NAV)
      COMPLEX V(1),CURR,YE,VA
C
      DO 10 J=1,NAV
      RHS(J)=0.
   10 CONTINUE
      IF (CURR.EQ.(0.,0.)) GO TO 100
      J2=0
      GO TO (20,30,40,50), LTP
   20 J1=LBINP(INL)
      J2=LBOUT(INL)
      GO TO 60
   30 J1=LBINP(INL)
      GO TO 60
   40 J1=LBOUT(INL)
      GO TO 60
   50 J1=INL
   60 VA=CONJG(CURR)*YE/CABS(CURR)
      I=0
      J=J1
   70 IF (BTYP(J).EQ.2) GO TO 90
      K=J+J
      IF (BTYP(J).EQ.1) GO TO 80
      RHS(K-1)=REAL(VA)
      RHS(K)=AIMAG(VA)
      GO TO 90
   80 RHS(K-1)=AIMAG(V(J)*VA)
   90 I=I+1
      IF (I.EQ.2.OR.J2.EQ.0) GO TO 100
      J=J2
      VA=-VA
      GO TO 70
  100 IF (IWRITE.LT.4) GO TO 110
      WRITE (6,120)
      WRITE (6,130) (I,RHS(I),I=1,NAV)
  110 RETURN
  120 FORMAT (//" RHS VECTOR OF ADJOINT EQUATIONS"/)
  130 FORMAT (5(2X,I4,":",E13.7))
      END
```

```
C
C
C
      SUBROUTINE RHST5 (LBINP,LBOUT,V,BTYP,RHS,YE,LTP,INL,NAV,IWRITE)
C
C     SUBROUTINE RHST5 FORMULATES RHS VECTOR OF THE ADJOINT EQUATIONS
C     FOR THE POWER LOSSES  OF THE INLTH ELEMENT. TYPE OF THE ELEMENT
C     IS IDENTIFIED BY THE PARAMETER LTP
C
      INTEGER LBINP(1),LBOUT(1),BTYP(1)
      REAL RHS(NAV)
      COMPLEX V(1),YE,VA
C
      DO 10 J=1,NAV
      RHS(J)=0.
   10 CONTINUE
      J2=0
      GO TO (20,30,40,50), LTP
   20 J1=LBINP(INL)
      J2=LBOUT(INL)
      VA=CONJG(V(J1)-V(J2))
      GO TO 60
   30 J1=LBINP(INL)
      VA=CONJG(V(J1))
      GO TO 60
   40 J1=LBOUT(INL)
      VA=CONJG(V(J1))
      GO TO 60
   50 VA=CONJG(V(INL))
   60 VA=2.*REAL(YE)*VA
      I=0
      J=J1
   70 IF (BTYP(J).EQ.2) GO TO 90
      K=J+J
      IF (BTYP(J).EQ.1) GO TO 80
      RHS(K-1)=REAL(VA)
      RHS(K)=AIMAG(VA)
      GO TO 90
   80 RHS(K-1)=AIMAG(V(J)*VA)
   90 I=I+1
      IF (I.EQ.2.OR.J2.EQ.0) GO TO 100
      J=J2
      VA=-VA
      GO TO 70
  100 IF (IWRITE.LT.4) GO TO 110
      WRITE (6,120)
      WRITE (6,130) (I,RHS(I),I=1,NAV)
  110 RETURN
  120 FORMAT (//" RHS VECTOR OF ADJOINT EQUATIONS"/)
  130 FORMAT (5(2X,I4,":",E13.7))
      END
```

```
C
C
C
      SUBROUTINE RHST6 (YT,JRYT,ICYT,V,BTYP,RHS,NAV,IWRITE)
C
C     SUBROUTINE RHST6 FORMULATES RHS VECTOR OF THE ADJOINT
C     EQUATIONS FOR THE TRANSMISSION POWER LOSSES
C
      INTEGER JRYT(1),ICYT(1),BTYP(1)
      REAL RHS(NAV)
      COMPLEX YT(1),V(1),Z
C
      N=NAV/2-1
      RHS(NAV-1)=0.
      RHS(NAV)=0.
      DO 30 I=1,N
      J1=JRYT(I)
      J2=JRYT(I+1)-1
      Z=(0.,0.)
      DO 10 J=J1,J2
      Z=Z+CONJG(V(ICYT(J)))*REAL(YT(J))
   10 CONTINUE
      Z=Z+Z
      K2=I+I
      K1=K2-1
      IF (BTYP(I).EQ.1) GO TO 20
      RHS(K1)=REAL(Z)
      RHS(K2)=AIMAG(Z)
      GO TO 30
   20 RHS(K1)=AIMAG(V(I)*Z)
      RHS(K2)=0.
   30 CONTINUE
      IF (IWRITE.LT.4) GO TO 40
      WRITE (6,50)
      WRITE (6,60) (I,RHS(I),I=1,NAV)
   40 RETURN
   50 FORMAT (//" RHS VECTOR OF ADJOINT EQUATIONS"/)
   60 FORMAT (5(2X,I4,":",E13.7))
      END
```

```
C
C
C
      SUBROUTINE RHST7 (YT,JRYT,ICYT,V,BTYP,RHS,NB,IWRITE)
C
C     SUBROUTINE RHST7 FORMS THE RIGHT HAND SIDE VECTOR RHS OF THE
C     ADJOINT EQUATIONS FOR THE ACTIVE POWER OF THE SLACK BUS
C
      INTEGER JRYT(1),ICYT(1),BTYP(NB)
      REAL RHS(1)
      COMPLEX YT(1),V(NB),Z,ZZ
C
      NAV=NB+NB
      NR=NAV-2
      DO 10 J=1,NR
      RHS(J)=0.0
   10 CONTINUE
      Z=CONJG(V(NB))
      RHS(NAV-1)=-REAL(Z)
      RHS(NAV)=-AIMAG(Z)
      K1=JRYT(NB)+1
      K2=JRYT(NB+1)-1
      DO 30 L=K1,K2
      IBUS=ICYT(L)
      LL=IBUS+IBUS
      ZZ=YT(L)*Z
      IF (BTYP(IBUS).EQ.1) GO TO 20
      RHS(LL-1)=REAL(ZZ)
      RHS(LL)=AIMAG(ZZ)
      GO TO 30
   20 RHS(L-1)=AIMAG(ZZ*V(IBUS))
   30 CONTINUE
      IF (IWRITE.LT.4) GO TO 40
      WRITE (6,50)
      WRITE (6,60) (I,RHS(I),I=1,NAV)
   40 RETURN
   50 FORMAT (//" RHS VECTOR OF ADJOINT EQUATIONS"/)
   60 FORMAT (5(2X,I4,":",E13.7))
      END
```

```
C
C
C
      SUBROUTINE RHST8 (YT,JRYT,ICYT,V,BTYP,RHS,CN,DN,PN,NAV,IWRITE)
C
C     SUBROUTINE RHST8 FORMULATES RHS VECTOR OF THE ADJOINT EQUATIONS
C     FOR THE COST OF GENERATION
C
      INTEGER JRYT(1),ICYT(1),BTYP(1)
      REAL RHS(NAV)
      COMPLEX YT(1),V(1),VA,VB
C
      NB=NAV/2
      VA=CONJG(V(NB))*(CN+2*DN*PN)
      K=NAV-2
      DO 10 J=1,K
      RHS(J)=0.
   10 CONTINUE
      RHS(NAV-1)=-REAL(VA)
      RHS(NAV)=-AIMAG(VA)
      J1=JRYT(NB)+1
      J2=JRYT(NB+1)-1
      DO 30 I=J1,J2
      VB=YT(I)*VA
      L=ICYT(I)
      K=L+L
      IF (BTYP(L).EQ.1) GO TO 20
      RHS(K-1)=REAL(VB)
      RHS(K)=AIMAG(VB)
      GO TO 30
   20 RHS(K-1)=AIMAG(V(I)*VB)
   30 CONTINUE
      IF (IWRITE.LT.4) GO TO 40
      WRITE (6,50)
      WRITE (6,60) (I,RHS(I),I=1,NAV)
   40 RETURN
   50 FORMAT (//" RHS VECTOR OF ADJOINT EQUATIONS"/)
   60 FORMAT (5(2X,I4,":",E13.7))
      END
```

```
C
C
C
      SUBROUTINE PDR3 (LBINP,LBOUT,VB,CCV,ICV,PDR,INB,NCV,NDR)
C
C
C     SUBROUTINE PDR3 CALCULATES PARTIAL DERIVATIVES OF A BUS REACTIVE
C     POWER W.R.T. LINE CURRENTS
C
      INTEGER LBINP(1),LBOUT(1),CCV(NCV),ICV(NCV)
      COMPLEX PDR(1),VB
      NDR=0
      DO 50 J=1,NCV
      L=CCV(J)
      M=ICV(J)
      IF (L.GT.6.AND.L.LT.11) GO TO 50
      NDR=NDR+1
      PDR(NDR)=(0.,0.)
      IF (L.GE.7) GO TO 30
      IF (L.GE.5) GO TO 20
      IF (L.GE.3) GO TO 10
      IF (LBINP(M).NE.INB) GO TO 50
      GO TO 40
   10 IF (LBINP(M).NE.INB.AND.LBOUT(M).NE.INB) GO TO 50
      GO TO 40
   20 IF (LBOUT(M).NE.INB) GO TO 50
      GO TO 40
   30 IF (M.NE.INB) GO TO 50
   40 PDR(NDR)=CMPLX(0.,-1.)*CONJG(VB)
   50 CONTINUE
      RETURN
      END
```

```
C
C
C
      SUBROUTINE PDR4 (CCV,ICV,PDR,CURR,LTP,INL,NCV,NDR)
C
C
C     SUBROUTINE PDR4 CALCULATES PARTIAL DERIVATIVES OF THE MODULUS
C     OF CURRENT OF THE INLTH ELEMENT W.R.T. LINE CURRENTS
C
      INTEGER CCV(NCV),ICV(NCV)
      COMPLEX PDR(1),CURR,VA
C
      NDR=0
      DO 10 J=1,NCV
      L=CCV(J)
      M=ICV(J)
      IF (L.GT.6.AND.L.LT.11) GO TO 10
      NDR=NDR+1
      PDR(NDR)=(0.,0.)
      IF (M.NE.INL) GO TO 10
      VA=-CONJG(CURR)/CABS(CURR)
      IF (L.LE.2.AND.LTP.EQ.2) PDR(NDR)=VA
      IF (L.EQ.3.OR.L.EQ.4.AND.LTP.EQ.1) PDR(NDR)=VA
      IF (L.EQ.5.OR.L.EQ.6.AND.LTP.EQ.3) PDR(NDR)=VA
      IF (L.EQ.11.AND.LTP.EQ.4) PDR(NDR)=VA
   10 CONTINUE
      RETURN
      END
```

```
C
C
C
      SUBROUTINE PDR5 (CCV,ICV,PDR,CURR,F,LTP,INL,NCV,NDR)
C
C     SUBROUTINE PDR5 CALCULATES PARTIAL DERIVATIVES OF THE POWER
C     LOSSES OF THE INLTH ELEMENT W.R.T. LINE CURRENTS
C
      INTEGER CCV(NCV),ICV(NCV)
      COMPLEX PDR(1),CURR,VA
      NDR=0
      DO 10 J=1,NCV
      L=CCV(J)
      M=ICV(J)
      IF (L.GT.6.AND.L.LT.11) GO TO 10
      NDR=NDR+1
      PDR(NDR)=(0.,0.)
      IF (M.NE.INL) GO TO 10
      VA=-2.*F/CURR
      IF (L.LE.2.AND.LTP.EQ.2) PDR(NDR)=VA
      IF (L.EQ.3.OR.L.EQ.4.AND.LTP.EQ.1) PDR(NDR)=VA
      IF (L.EQ.5.OR.L.EQ.6.AND.LTP.EQ.3) PDR(NDR)=VA
      IF (L.EQ.11.AND.LTP.EQ.4) PDR(NDR)=VA
   10 CONTINUE
      RETURN
      END
```

```
C
C
C
      SUBROUTINE PDR6 (LBINP,LBOUT,LINPG,LINPB,LG,LB,V,CCV,ICV,PDR,NCV,N
     1DR)
C
C     SUBROUTINE PDR6 CALCULATES PARTIAL DERIVATIVES OF TRANSMISSION
C     POWER LOSSES W.R.T. LINE CURRENTS
C
      INTEGER LBINP(1),LBOUT(1),CCV(1),ICV(1)
      REAL LINPG(1),LINPB(1),LG(1),LB(1)
      COMPLEX PDR(1),V(1)
C
      NDR=0
      DO 10 J=1,NCV
      L=CCV(J)
      M=ICV(J)
      IF (L.GE.7.AND.L.NE.11) GO TO 10
      NDR=NDR+1
      IF (L.LE.2) PDR(NDR)=-2.*CONJG(V(LBINP(M)))*LINPG(M)/CMPLX(LINPG(M
     1),LINPB(M))
      IF (L.EQ.3.OR.L.EQ.4) PDR(NDR)=-2.*CONJG(V(LBINP(M))-V(LBOUT(M)))*
     1LG(M)/CMPLX(LG(M),LB(M))
      IF (L.EQ.5.OR.L.EQ.6) PDR(NDR)=-2.*CONJG(V(LBOUT(M)))*LINPG(M)/CMP
     1LX(LINPG(M),LINPB(M))
      IF (L.EQ.11) PDR(NDR)=(0.,0.)
   10 CONTINUE
      RETURN
      END
```

```
C
C
C
      SUBROUTINE PDR7 (LBINP,LBOUT,VB,CCV,ICV,PDR,INB,NCV,NDR)
C
C     SUBROUTINE PDR7 CALCULATES PARTIAL DERIVATIVES OF A BUS ACTIVE
C     POWER W.R.T. LINE CURRENTS
C
      INTEGER LBINP(1),LBOUT(1),CCV(NCV),ICV(NCV)
      COMPLEX PDR(1),VB
C
      NDR=0
      DO 50 J=1,NCV
      L=CCV(J)
      M=ICV(J)
      IF (L.GT.6.AND.L.LT.11) GO TO 50
      NDR=NDR+1
      PDR(NDR)=(0.,0.)
      IF (L.GE.7) GO TO 30
      IF (L.GE.5) GO TO 20
      IF (L.GE.3) GO TO 10
      IF (LBINP(M).NE.INB) GO TO 50
      GO TO 40
   10 IF (LBINP(M).NE.INB.AND.LBOUT(M).NE.INB) GO TO 50
      GO TO 40
   20 IF (LBOUT(M).NE.INB) GO TO 50
      GO TO 40
   30 IF (M.NE.INB) GO TO 50
   40 PDR(NDR)=-CONJG(VB)
   50 CONTINUE
      RETURN
      END
```

```
C
C
C
      SUBROUTINE PDR3 (LBINP,LBOUT,VB,CCV,ICV,PDR,CN,DN,PN,NB,NCV,NDR)
C
C     SUBROUTINE PDR3 DETERMINES PARTIAL DERIVATIVES OF THE COST OF
C     GENERATION W.R.T. LINE CURRENTS
C
      INTEGER LBINP(1),LBOUT(1),CCV(NCV),ICV(NCV)
      COMPLEX PDR(1),VB
C
      NDR=0
      DO 50 J=1,NCV
      L=CCV(J)
      M=ICV(J)
      IF (L.GT.6.AND.L.LT.11) GO TO 50
      NDR=NDR+1
      PDR(NDR)=(0.,0.)
      IF (L.GE.7) GO TO 30
      IF (L.GE.5) GO TO 20
      IF (L.GE.3) GO TO 10
      IF (LBINP(M).NE.NB) GO TO 50
      GO TO 40
   10 IF (LBINP(M).NE.NB.AND.LBOUT(M).NE.NB) GO TO 50
      GO TO 40
   20 IF (LBOUT(M).NE.NB) GO TO 50
      GO TO 40
   30 IF (M.NE.NB) GO TO 50
   40 PDR(NDR)=-(CN+2.0*DN*PN)*CONJG(VB)
   50 CONTINUE
      RETURN
      END
```

```
C
C
C
      SUBROUTINE WRTGR (CCV,ICV,DF,NCV)
C
C     SUBROUTINE WRTGR WRITES GRADIENTS OF A FUNCTION
C
      INTEGER CCV(NCV),ICV(NCV)
      REAL DF(NCV),RLB(11)
      DATA RLB/"LINPG(","LINPB(","    LG(","    LB(","LOUTG(","LOUTB(","
     1  P(","     Q(","MOD V(","ARG V("," BSTL("/
      WRITE (6,10) (RLB(CCV(J)),ICV(J),DF(J),J=1,NCV)
   10 FORMAT ((1X,4(3X,A6,I3,"):",1X,E13.7)))
      END
```

```
C
C
C
      COMPLEX FUNCTIONCF1(C0,Z1,Z2,LST1,LST2,N,IDAT)
C
C     FUNCTION  SUBPROGRAM  CF1 CALCULATES THE VALUE OF A COMPLEX
C     FUNCTION
C     CF1=C0 + Z1(LST1(1))*Z2(LST2(1)) + Z1(LST1(2))*Z2(LST2(2)) + ..
C         .. + Z1(LST1(N))*Z2(LST2(N))
C     OF COMPLEX ARGUMENTS
C
      INTEGER LST1(1),LST2(1)
      COMPLEX Z1(1),Z2(1),C0
C
      CF1=C0
      IF (IDAT.GE.1) GO TO 20
      DO 10 K=1,N
      CF1=CF1+Z1(K)*Z2(K)
   10 CONTINUE
      RETURN
   20 IF (IDAT.GE.2) GO TO 40
      DO 30 K=1,N
      CF1=CF1+Z1(K)*Z2(LST2(K))
   30 CONTINUE
      RETURN
   40 IF (IDAT.GE.3) GO TO 60
      DO 50 K=1,N
      CF1=CF1+Z1(LST1(K))*Z2(K)
   50 CONTINUE
      RETURN
   60 DO 70 K=1,N
      CF1=CF1+Z1(LST1(K))*Z2(LST2(K))
   70 CONTINUE
      RETURN
      END
```

```
C
C
C
      COMPLEX FUNCTIONCF5(C0,Z1,Z2,LST1,LST2,N,IDAT)
C
C     FUNCTION SUBPROGRAM CF5 CALCULATES THE VALUE OF A COMPLEX
C     FUNCTION
C     CF5=C0*CONJG(CF1((0.,0),Z1,Z2,LST1,LST2,N,IDAT))
C
      INTEGER LST1(1),LST2(1)
      COMPLEX Z1(1),Z2(1),C0,CF1
      CF5=C0*CONJG(CF1((0.,0.),Z1,Z2,LST1,LST2,N,IDAT))
      RETURN
      END
```